J. Nig. Ass. Math. Phys.
Vol. 1 (1992) 71 - 75

A BASIC CODE FOR PLOTTING DISPERSION CURVES IN A COLD PLASMA

by

S. D. Salihu & B. Chike-Obi
Department of Physics
University of Ilorin
Ilorin, Nigeria.

ABSTRACT
We have developed a BASIC code for plotting the dispersion curves for waves in a cold plasma. The programme algorithm is based on the evaluation of a many-valued polynomial function of wave frequency and angle of propagation. Typical dispersion curves are discussed. Design efforts to minimize the effects of singularities on the curves are analyzed.

1. INTRODUCTION

We have plotted the dispersion curves for the propagation of electromagnetic waves in a cold, uniform, and magnetized plasma, in order to obtain the cutoffs and resonances of these waves, in other words, to determine the electromagnetic wave frequencies and wavelengths which it is possible to find in this type of plasma. The analysis of a cold uniform plasma should be extended to include the more realistic model of a warm inhomogeneous plasma. Nevertheless, the results which are obtained in the cold plasma approximation are very close to measurements and to theoretical results obtained with more realistic models. Our main motivation for developing the BASIC code to be discussed in this paper, is that we would like to reproduce known results for electromagnetic wave cutoffs and resonances in a cold uniform plasma. Secondly, our code is the first step in an effort to develop a BASIC code which can search for roots, and, ultimately, solve boundary layer ordinary differential equations of the type which are encountered in many of the problems in theoretical plasma physics. Codes of this type are available in FORTRAN, but our work is partly motivated by their scarcity in BASIC, a programme language widely available on personal computers, which are more common in Nigeria.

2. THE DISPERSION RELATION

In a cold uniform plasma immersed in a uniform static magnetic field and a uniform time-varying electric field, the dielectric tensor $\varepsilon_{\alpha\beta}$ is obtained from the equation of motion of the charged particles. For the electrons, the equation of motion is

$$m d\underline{v}/dt = -e\underline{E} - e\underline{v} \times \underline{B}_o/c \qquad (1)$$

Assuming that all the time-varying quantities vary as $\exp(-i\omega t)$, we have, from eq (1),

$$i\omega \underline{v} + e\underline{B}_o \times \underline{v}/mc = e\underline{E}/m \qquad (2)$$

To solve for $\underline{v}$, we multiply eq (2) by the conjugate operator $(-i\omega$

$+ (eB_o/mc)\, x$) to find

$$\underline{v} = -(ie\omega/m(\omega^2 - \omega_{Be}^2))(\underline{E} - \omega_{Be}\underline{b}(\underline{E}\cdot\underline{b})/\omega^2 - (i\omega_{Be}\underline{E}\times\underline{b})/\omega) \quad (3a)$$

where $\underline{b} = \underline{B}_o/B$ and $\omega_{Be} = eB_o/mc$. The components of $\underline{v}$ are

$$v_\alpha = \nu_{\alpha\beta} E_\beta \quad (3b)$$

where $\nu_{\alpha\beta}$ is the mobility tensor. The plasma polarization $\underline{P}$ and induction $\underline{D}$, due to electron motion, are related to velocity $\underline{v}$ by $-i\omega\underline{P} = -i\omega(\underline{D} + \underline{E})/4\pi = \underline{J} = en_e\underline{v}$.
In tensor form, $D_\alpha = \varepsilon_{\alpha\beta}E_\beta$. From this,

$$\varepsilon_{\alpha\beta} = 1 - 4\pi e n_e \nu_{\alpha\beta}/i\omega \quad (4)$$

The ion contribution to polarization is calculated in the same way, and with the electron contribution, we obtain the result

$$\varepsilon_{\alpha\beta} = \begin{pmatrix} \varepsilon & -ig & 0 \\ ig & \varepsilon & 0 \\ 0 & 0 & \eta \end{pmatrix} \quad (5)$$

where

$$\varepsilon_{xx} = \varepsilon_{yy} \equiv \varepsilon = 1 - \omega_{pe}^2(\omega^2 - \omega_{Be}^2)^{-1} - \omega_{pi}^2(\omega^2 - \omega_{Bi}^2)^{-1} \quad (6a)$$

$$\varepsilon_{zz} \equiv \eta = 1 - (\omega_{pe}^2 + \omega_{pi}^2)/\omega^2 \quad (6b)$$

and

$$-i\varepsilon_{xy} = g = \omega_{Be}\omega_{pe}^2\omega^{-1}(\omega^2 - \omega_{Be}^2)^{-1} - \omega_{Bi}\omega_{pi}^2\omega^{-1}(\omega^2 - \omega_{pi}^2)^{-1} \quad (6c)$$

In eq (6), the components of $\varepsilon_{\alpha\beta}$ are given under the assumption that the z-axis is along $\underline{B}_o$. The cyclotron and plasma frequencies for the electrons and ions are $\omega_{Be} = eB_o/mc$, $\omega_{Bi} = ZeB_o/Mc$, $\omega_{pe} = (4\pi n_e e^2/m)^{\frac{1}{2}}$, and $\omega_{pi} = (4\pi n_i e^2/M)^{\frac{1}{2}}$. For an electromagnetic field with time and space dependence of the form $\exp(-i\omega t + i\underline{k}\cdot\underline{r})$, Maxwell's equations give

$$\underline{k}\times\underline{E} = \omega\underline{B}/c \quad (7a)$$

$$\underline{k}\times\underline{B} = -i\underline{D}/c \quad (7b)$$

These equations give

$$\omega^2\underline{D}/c^2 = -\underline{k}\times(\underline{k}\times\underline{E}) = \underline{E}k^2 - \underline{k}(\underline{k}\cdot\underline{E})$$

or, in components,

$$E_\alpha k^2 - k_\alpha k_\beta E_\beta = \omega^2 D_\alpha/c^2 = \omega^2\varepsilon_{\alpha\beta}E_\beta/c^2 \quad (8)$$

This system of equations has nontrivial solutions only if

$$\mathrm{Det}\left[k^2\delta_{\alpha\beta} - k_\alpha k_\beta - \omega^2\varepsilon_{\alpha\beta}/c^2\right] = 0 \quad (9)$$

Eq (9), which relates ω and $\underline{k}$, is the desired dispersion relation. Recalling the definition of the refractive index N, $kc/\omega \equiv N$, and choosing $\underline{k}$ to be in the x-z plane, so that

$$\underline{k} = (k\sin(\theta),\ 0,\ k\cos(\theta)),$$

where θ is the angle between $\underline{k}$ and $\underline{B}_o$, we obtain from equations (5) and (9) the expression

$$AN^4 + BN^2 + C = 0 \qquad (10)$$

where

$$A = \varepsilon\sin^2(\theta) + \eta\cos^2(\theta),$$

$$B = -\varepsilon\eta(1 + \cos^2(\theta)) - (\varepsilon^2 - g^2)\sin^2(\theta),$$

and

$$C = \eta(\varepsilon^2 - g^2).$$

Solving the quadratic equation (10), we find (see Shafranov, 1964),

$$N_{\pm}^2 = \left\{(\varepsilon^2 - g^2 - \varepsilon\eta)\sin^2(\theta) + 2\varepsilon\eta \pm \left[(\varepsilon^2 - g^2 - \varepsilon\eta)^2\sin^4(\theta) + 4\eta^2 g^2\cos^2(\theta)\right]^{\frac{1}{2}}\right\}/2A \qquad (11)$$

3. DISCUSSION OF THE DISPERSION CURVES

In order to plot $N^2(\omega)$, one requires the values of ω_{Be}, ω_{Bi}, ω_{pi}, and θ. We have chosen the following values for the purpose of illustration and clarity of the graphs. The ratio of the masses, M/m, is taken to be 4, although the actual value of this ratio is about 2000. The value of ω_{Bi} is taken to be 1, and the ω-axis is divided in multiples of ω_{Bi}. The ratio ω_{Be}/ω_{pe} is taken to be 2. The normalized values then become $\omega_{Bi} = 1$, $\omega_{Be} = 4$, $\omega_{pe} = 2$, and $\omega_{pi} = 1$. In our code, the ω-axis starts from 0.05 rather than zero, in order to avoid the infinities in eq (6). Another possible infinity in eq (6) is avoided by choosing an interval size 0.105, so that successive points increase by more than one decimal point, and $\omega/\omega_{Bi} \neq 1$. The standard practice (Kemeny & Kurtz, 1980) is to define the range of $N^2(\omega)$ in the programme, before computation, in order to avoid undesired high values which might occur near the resonances. It is also important to note, from eq (11), that there are two values of N^2 for each value of ω. The two graphs which result from this interchange with one another whenever they cross the infinity points. Figure 1 shows the plot for $\theta = 30^\circ$. There are five branches in all, agreeing with theory (see Lifshitz & Pitaevskii, 1981). The resonant frequencies, $N^2 = \infty$, are read directly from the graphs to be

$$\begin{aligned} \omega_{r1} &= 4.15\omega_{Bi} \\ \omega_{r2} &= 2.05\omega_{Bi} \\ \omega_{r3} &= 1.94\omega_{Bi} \end{aligned} \qquad (12)$$

which are the roots of eq (10), the dispersion equation. The cut-off frequencies, $N^2 = 0$, are

$$\begin{aligned} \omega_{o1} &= 4.88\omega_{Bi} \\ \omega_{o2} &= 1.84\omega_{Bi} \end{aligned} \qquad (13)$$

Figure 2 shows the plots for $\theta = 0$, 30°, 60°, and 90°. The general configuration of the graphs is the same as in figure 1, since the ratio ω_{Be}/ω_{pe} is constant for all the angles. The fifth branch, which occurs at very low frequencies, disappears in the case $\theta = 60^\circ$ because the same scale is used for all the plots. A wider scale would have displayed this branch. However, there are four branches for $\theta = 0$ and 90° as expected, and the lower and upper hybrid frequencies are found to be

$$\omega_{lh} = 1.21\omega_{Bi}$$

$$\omega_{uh} = 4.36\omega_{Bi}.$$



Figure 1



Figure 2

4. CONCLUSIONS

There are singularities in the dispersion curves for electromagnetic waves in a cold plasma. The various ways by which these infinities may be avoided in plotting the curves were analyzed. The resulting BASIC code (listed in the appendix) incorporates techniques to avoid the singularities. It is possible to read off the resonant and cut-off frequencies, but the accuracy of the results depends on the range of N^2 used, and the intervals of ω. In the examples given in this paper, a range of -2.5 to 6.0 for N^2, and an interval of 0.105 were used. The results obtained for resonances and cutoffs (see equations (12) and (13)) are accurate to within $0.05\omega_{Bi}$. These results show that our code is a simple and effective means of plotting plasma dispersion curves, its added advantage being that it is in BASIC because this widens the range of computers in which it can be implemented.

APPENDIX

```
10 REM REFRACTIVE INDEX OF PLASMA
20 REM THIS PROGRAMME PLOTS THE SQUARE OF THE
30 REM REFRACTIVE INDEX N AGAINST THE FREQUENCY W
40 REM FOR ANY ANGLE O
50 REM B(1),B(2) ARE THE CYCLOTRON FREQUENCIES
60 REM OF THE ELECTRONS AND THE IONS RESPECTIVELY
70 REM P(1),P(2) ARE THEIR PLASMA FREQUENCIES
80 REM P(2) IS SET EQUAL TO 1 FOR NORMALIZATION
90 REM FOR W RUNNING FROM A TO B IN STEPS OF S
100 REM THE RANGE OF N IS FROM C TO D
110 REM SPREAD OVER 70 CHARACTER POSITIONS
120 REM THE CHARACTER POSITIONS AT THE LEFT MARGIN IS 8
130 READ A,B,S,C,D,A1,B1,B(1),B(2),P(1),P(2),O
140 DATA .05,6,.105,-2.5,4.5,70,8,4,1,2,1,1.0472
150 REM --- LABELLING OF Y-AXIS ---
160 H=(D-C)/70
170 PRINT "N-AXIS:FROM";C;"TO";D;"IN STEP OF";H
180 PRINT TAB(8);".";
190 FOR I = 1 TO 69
200 PRINT "-";
210 NEXT I
220 PRINT "."
230 REM -------------------- BREAK UP OF THE EQUATION ---
240 FOR W = A TO B STEP S
250 Q=1-P(1)^2/(W^2-B(1)^2)-P(2)^2/(W^2-1)
260 R=(B(1)*P(1)^2)/(W^3-W*B(1)^2)-P(2)^2/(W^3-W)
270 T=1-(P(1)^2+P(2)^2)/W^2
280 U=(Q^2-R^2-Q*T)*SIN(O)^2+2*Q*T
290 V=((Q^2-R^2-Q*T)^2)*SIN(O)^4+4*T^2*R^2*COS(O)^2
300 X=SQR(V)
310 Z=2*(Q*SIN(O)^2+T*COS(O)^2)
320 REM -------------------- FINAL FORMULAE ----------------
330 DEF FNF(W)=(U+X)/Z
340 DEF FNG(W)=(U-X)/Z
350 REM -------------------- PLOTTING THE GRAPH ------------
360 DEF FNR(W)=INT(W+.5)
370 DEF FNM(W)=INT(100*W+.5)/100
380 N=FNF(W)
390 N1=A1+FNR((N-C)/H)
400 N=FNG(W)
410 N2=A1+FNR((N-C)/H)
420 REM ----------------- NEGLECTING VALUES OUTSIDE RANGE ---------
430 IF N1<8 THEN 520
440 IF N1>78 THEN 550
450 IF N2<8 THEN 580
460 IF N2>78 THEN 580
470 IF N2<N1 THEN 500
480 PRINT FNM(W);TAB(N1);"*";TAB(N2);"*"
490 GOTO 630
500 PRINT FNM(W);TAB(N2);"*";TAB(N1);"*"
510 GOTO 630
520 IF N2<8 THEN 620
530 IF N2>78 THEN 620
540 GOTO 600
550 IF N2<8 THEN 620
560 IF N2>78 THEN 620
570 GOTO 600
580 PRINT FNM(W);TAB(N1);"*"
590 GOTO 630
600 PRINT FNM(W);TAB(N2);"*"
610 GOTO 630
620 PRINT FNM(W)
630 NEXT W
640 END
```

REFERENCES

J. G. Kenemy & T. E. Kurtz, (1980) "BASIC programming", John Wiley

E. M. Lifshitz & L. P. Pitaevskii, (1981) "Physical kinetics", Pergamon

V. D. Shafranov, (1967) in "Reviews of plasma physics", 2, (M. A. Leontovich, Ed.), Consultants Bureau, New York